
 17.10.2002 – 1.00גירסה 
 

 הגבלת הגולשים באתר
 

 . il.co.livedns.underwar://httpמסמך זה הורד מהאתר 
 .ללא אישור מפורש מאת המחברים, אין להפיץ מסמך זה במדיה כלשהי

עקב השימוש במידע שיגרם , ישיר או עקיף, מחברי המסמך אינם אחראים לכל נזק
, עם זאת. וכן לנכונות התוכן של הנושאים המופיעים במסמך, המופיע במסמך

 .המחברים עשו את מירב המאמצים כדי לספק את המידע המדויק והמלא ביותר
 

  ואסף רשףניר אדרכל הזכויות שמורות ל
 
Nir Adar 
Email: underwar@hotmail.com  
Home Page: http://underwar.livedns.co.il 
 
Assaf Reshef 
Email: assaf@fullscreen.co.il 
 

 
 

 .יםאנא שלחו תיקונים והערות אל המחבר

 הקדמה
 

הוא צריך לדאוג גם , ן למבקרים גישה לראות את קוד המקור של דפי האתר שלוכאשר בעל אתר נות
 . אחרת יהיה אפשר לנצל את זה לרעה, לאבטחה נכונה של הקוד שמציג אותם
 : פרסם מסמך בנושאL0phtאחד מחברי , 1999, הבעיה ידועה זמן רב ובשביעי במאי

"Web users can view ASP source code and other sensitive files on the web server" ,
 .בעיה זו עדיין אקטואלית, אפילו בימים אלו, עם זאת. אשר הציגה את הבעיה בפירוט
 .אולם ניתן למצוא את אותה הבעיה ובעיות נוספות באתרים רבים, נדגים בעיה זו על אתר אחד

 
ת הדוגמאות המצורפת על והוא חלק מחביל, showcode.aspהסקריפט הבעייתי שנציג במסמך זה הוא 

Microsoft IIS Server. 

 מציאת אתר עם בעיה
 
 

 .asp.5request/example/asp06/webprog/books/jang/tw.edu.nthu.cs.neural://http: נתחיל בביקור בדף 
 
 .ASP שימושי זה מאפשר לגולש לראות את משתני השרת השונים שקיימים בשפת דף
 

 
 
 

 .View Source –ישנו קישור מעניין , נשים לב שבתחתית הדף
הדף המגיע אלינו לדפדפן אינו הדף , כלומר,  הינם דפים העוברים עיבוד בצד השרתASPנזכיר כי דפי 

אם נרצה לאפשר למבקרים לראות את דפי ה. חרי עיבודאלא אנו מקבלים דף א, המקורי שהמתכנת כתב
ASPשיהיה גם הוא דף , נצטרך לכתוב סקריפט מיוחד,  המקורייםASPושידאג ,  שיישב בצד השרת

, View Sourceכאשר לחצנו על הקישור . זהו המקרה באתר שלפנינו. ASP-להצגת הקוד של דפי ה
עם הכתובת , נפתח חלון חדש

/webprog/books/jang=/asp?source.showcode/common/webprog/books/jang/tw.edu.nthu.cs.neural://http
asp.5request/example/asp06 ,נו הקוד של הדףשם נגלה לפני. 

 ?האם זוהי בעיית אבטחה?  האם אנו יכולים לנצל את הדף–חקירת הבעיה 
 

  asp.showcode/common/webprog/books/jang/tw.edu.nthu.cs.neural://httpננסה לנתח את הכתובת 
: אנו רואים כי הוא מקבל פרמטר. זהו הדף שמציג את הקוד

source=/jang/books/webprog/06asp/example/request5.asp. 
 .ניתן להבין כי פרמטר זה קובע את הקוד של הדף שיוצג בפנינו

 
אדם בעל גישה . נניח כי באתר שלנו יש מערכת משתמשים: הצגת הקוד של דפים פותחת בעיית אבטחה

להרוס /ולגנוב,  המקור יוכל לגלות היכן מצוי מבנה הנתונים השומר את כל הנתונים של המשתמשיםלקוד
 .במידה ויש לאדם את הקוד של האתר, בעיות אבטחה נוספות עלולות להתגלות. אותו
, לדוגמא, או לאפשר להציג רק דפים בתיקיה מסוימת, בדרך כלל נרצה למנוע את הצגת הדפים, לפיכך

 jang/books/webprog/נוכל לראות רק את הקוד שתחת הספרייה , אתר שאנו בודקים כעתייתכן שב
 .ואת כל תתי הספריות שלה

 .נבדוק זאת
: על מנת להציג את הדף , ונשנה אותו,  המקוריURL-ניקח את ה

asp.index/jang/tw.edu.nthu.cs.neural://http  
 :נכתוב בדפדפן את הכתובת הבאה

http://neural.cs.nthu.edu.tw/jang/books/webprog/common/showcode.asp?source=/jang/index.asp 
 

 :אנו מקבלים מולנו את הדף הבא, כן חסם את הגישה לכתובת זובעל האתר א

 .עדיין איננו מוותרים על ההזדמנות להביט בדפי האתר
 :ננסה לבדוק האם אנו יכולים לראות את הקוד של דף אחר

http://neural.cs.nthu.edu.tw/jang/books/webprog/common/showcode.asp?source=/jang/books/webprog/
index.asp. 

 
 . asp.index/webprog/books/jang/tw.edu.nthu.cs.neural://httpעל המסך מופיע הקוד של הדף 

 

 
 

 .היא ספרייה אחת כלפי מעלה".." המשמעות של הביטוי , ידוע כי במערכות הפעלה שונות
: ננסה כעת לראות את הקוד של האתר הבא

asp.index/../../webprog/books/jang/tw.edu.thun.cs.neural://http ,אותה כתובת , כלומר את הכתובת הבאה
 asp.index/jang/tw.edu.nthu.cs.neural://http: שקודם נחסמה אליה הגישה

 :נכתוב בדפדפן את השורה הבאה
 

http://neural.cs.nthu.edu.tw/jang/books/webprog/common/showcode.asp?source=/jang/books/webprog/
../../index.asp  

 

 
 

 .והקוד של הדף הוצג לפנינו ללא כל מחסום, הפעם ההגנה של השרת לא פעלה, אכן

 שימושים
 

 .כעת לנסות להשיג דברים נוספיםנוכל 
 :Perlניתן לראות בדף כי כותב האתר משתמש בסקריפט של , למשל

 
<table align=right border=3> 
<td><a href="sandbox/html/autoload.htm" target="new"><img src="/jang/cgi-
bin/randroge.pl" border=0></a> 
</table> 
 

 ?קוד של הסקריפטהאם אנו יכולים להציג את ה
 :ננסה

 
http://neural.cs.nthu.edu.tw/jang/books/webprog/common/showcode.asp?source=/jang/books/webprog/
../../cgi-bin/randroge.pl 

 
 .הסקריפט מוצג ללא כל בעיה

 
 :נראה האם נוכל למצוא את בעיית האבטחה. שמציג את כל הדפים, כעת ננסה להציג את הקוד של הדף

 :נביט בכתובת
http://neural.cs.nthu.edu.tw/jang/books/webprog/common/showcode.asp?source=/jang/books/webprog/
common/showcode.asp 
 

 .על ידי ניסוי וטעייה כפי שעשינו, ל"שלא היינו צריכים לגלות את בעיית האבטחה הנ, נשים לב
ולכן יכלנו , ".."בלי להשתמש ב ,  באופן חופשיshowcode.aspד של הדף אנו מסוגלים לראות את הקו

 וחיפוש בעיות showcode.aspלגלות את בעיית האבטחה שהרגע גילינו על ידי הסתכלות בקוד של 
 . אבטחה בקוד

 
 .נעשה זאת כעת



 אסף רשף  ניר אדר
com.underwar@hotmail  il.co.assaf@fullscreen 

UnderWarrior 2002 Team  http://underwar.livedns.co.il 
 

 8עמוד 
 

 חקירת הסיבות לבעיה
 

 :נביט בקוד של הדף
 

<SCRIPT  LANGUAGE=VBScript RUNAT=Server> 
FUNCTION fValidPath (ByVal strPath) 
      If (InStr(1, strPath, "/webbook/javascript", 1) or _ 
           InStr(1, strPath, "/books/webprog", 1)) Then 
           fValidPath = 1 
      Else 
           fValidPath = 0 
      End If 
END FUNCTION 
 
REM Returns the minimum number greater than 0 
REM If both are 0, returns -1 
FUNCTION posMin (iNum1, iNum2) 
      If iNum1 = 0 AND iNum2 = 0 Then 
           posMin = -1 
      ElseIf iNum2 = 0 Then 
           posMin = iNum1 
      ElseIf iNum1 = 0 Then 
           posMin = iNum2 
      ElseIf iNum1 <  iNum2 Then 
           posMin = iNum1 
      Else  
           posMin = iNum2 
      End If 
END FUNCTION 
 
FUNCTION fCheckLine(ByVal strLine) 
      fCheckLine = 0 
      iTemp = 0 
 
      iPos = InStr(strLine, "<" & "%") 
      If posMin(iTemp, iPos) = iPos Then 
           iTemp = iPos 
           fCheckLine = 1 
      End If 
    
      iPos = InStr(strLine, "%" & ">") 
      If posMin(iTemp, iPos) = iPos Then 
           iTemp = iPos 
           fCheckLine = 2 
      End If 
 
      iPos = InStr(1, strLine, "<" & "SCRIPT", 1) 
      If posMin(iTemp, iPos) = iPos Then 
           iTemp = iPos 
           fCheckLine = 3 
      End If 
 
      iPos = InStr(1, strLine, "<" & "/SCRIPT", 1) 
      If posMin(iTemp, iPos) = iPos Then 
           iTemp = iPos 

           fCheckLine = 4 
      End If 
END FUNCTION 
 
SUB PrintHTML(ByVal strLine) 
      iSpaces = Len(strLine) - Len(LTrim(strLine)) 
      i = 1 
      While Mid(Strline, i, 1) = Chr(9) 
           iSpaces = iSpaces + 5 
           i = i + 1 
      Wend 
      If iSpaces > 0 Then 
           For i = 1 to iSpaces 
                Response.Write(" ") 
           Next 
      End If 
      iPos = InStr(strLine, "<") 
      If iPos Then 
           Response.Write(Left(strLine, iPos - 1)) 
           Response.Write("<") 
           strLine = Right(strLine, Len(strLine) - iPos) 
           Call PrintHTML(strLine) 
      Else 
           Response.Write(strLine) 
      End If 
END SUB 
       
SUB PrintLine (ByVal strLine, iFlag) 
      Select Case iFlag 
           Case 0 
                Call PrintHTML(strLine) 
           Case 1 
                iPos = InStr(strLine, "<" & "%") 
                Call PrintHTML(Left(strLine, iPos - 1)) 
                Response.Write("<FONT COLOR=#ff0000>") 
                Response.Write("<%") 
                strLine = Right(strLine, Len(strLine) - (iPos + 1)) 
                Call PrintLine(strLine, fCheckLine(strLine)) 
           Case 2 
                iPos = InStr(strLine, "%" & ">") 
                Call PrintHTML(Left(strLine, iPos -1)) 
                Response.Write("%>") 
                Response.Write("</FONT>") 
                strLine = Right(strLine, Len(strLine) - (iPos + 1)) 
                Call PrintLine(strLine, fCheckLine(strLine)) 
           Case 3 
                iPos = InStr(1, strLine, "<" & "SCRIPT", 1) 
                Call PrintHTML(Left(strLine, iPos - 1)) 
                Response.Write("<FONT COLOR=#0000ff>") 
                Response.Write("<SCRIPT") 
                strLine = Right(strLine, Len(strLine) - (iPos + 6)) 
                Call PrintLine(strLine, fCheckLine(strLine)) 
           Case 4 
                iPos = InStr(1, strLine, "<" & "/SCRIPT>", 1) 
                Call PrintHTML(Left(strLine, iPos - 1)) 

                Response.Write("</SCRIPT>") 
                Response.Write("</FONT>") 
                strLine = Right(strLine, Len(strLine) - (iPos + 8)) 
                Call PrintLine(strLine, fCheckLine(strLine)) 
           Case Else 
                Response.Write("FUNCTION ERROR -- CONTACT ADMIN.") 
      End Select 
END SUB 
</SCRIPT> 
 
<%  strVirtualPath=Request("source") %> 
 
<html> 
<head><title>View Page Source</title></head> 
<body> 
 
<center> 
<font face="Verdana, Arial, Helvetica" SIZE=6>View Source</font> 
</center> 
 
<font FACE="Verdana, Arial, Helvetica" SIZE=2> 
Go Back to <a href="<%=strVirtualPath%>"><%=strVirtualPath%></a><BR> 
Go Back to <a href="/jang/sandbox">WWW sandbox</a> 
<hr> 
<% 
If fValidPath(strVirtualPath) Then 
      strFilename = Server.MapPath(strVirtualPath) 
      Set FileObject = Server.CreateObject("Scripting.FileSystemObject") 
      Set oInStream = FileObject.OpenTextFile (strFilename, 1, FALSE) 
      While NOT oInStream.AtEndOfStream 
           strOutput = oInStream.ReadLine 
           Call PrintLine(strOutput, fCheckLine(strOutput)) 
           Response.Write("<BR>") 
      Wend 
Else 
      Response.Write("<H1>View Source -- Access Denied</H1>") 
End If  
%> 
 
</body> 
</html> 
 

פעולת הסקריפט . בהם משתמש בדף,  החלק הכחול הוא פונקציות עזר–הדף מחולק למספר קטעים 
 .מתחילה בסוף הקטע הכחול

 . ממנו נתחיל את הניתוח–הבלוק האדום בסוף הדף . HTMLהקטע שאינו מודגש הינו קוד 
ם להציג למשתמש קובע הא, ולפי הערך שהיא מחזירה, ()fValidPathנראה כי המתכנת קורא לפונקציה 

 .View Source - Access Denied –או את ההודעה , את הדף המבוקש

 :זוהי הפונקציה האחראית לאבטחה בדף זה,()fValidPathנביט בעיון בקוד של 
 

FUNCTION fValidPath (ByVal strPath) 
      If (InStr(1, strPath, "/webbook/javascript", 1) or _ 
           InStr(1, strPath, "/books/webprog", 1)) Then 
           fValidPath = 1 
      Else 
           fValidPath = 0 
      End If 
END FUNCTION 

 
 .הפונקציה מקבלת מחרוזת המציינת נתיב בשרת

 books/webprog/ או הכתובת webbook/javascript/כותב הסקריפט התכוון לבדוק אם הכתובת 
לכן "..". על ידי , בנתיבים" דילוג אחורה"אבל הוא לא חשב על האפשרות של , URL-עות במופי

בנתיב שחיפשנו גרמו לנו ".." וה , הפונקציה אישרה לנו את זה, כשהכנסנו את הטקסט שהוא רצה לראות
 בעצם לבקש את הקוד של סקריפט שהכותב לא התכוונן שנראה

 
 
 

 פתרון אפשרי לבעיה
 

 : היה יכול להיות למשלפתרון לבעיה
 
 

FUNCTION fValidPath (ByVal strPath) 
      If (InStr(1, strPath, "/webbook/javascript", 1) or _ 
           InStr(1, strPath, "/books/webprog", 1)) Then 
           fValidPath = 1 
      Else 
           fValidPath = 0 
      End If 
      If (InStr(strPath, “..”) <> 0) Then fValidPath = 0 
END FUNCTION 

 
בנתיב שלנו ולא מאפשרת לקוד להראות קבצים עם הטקסט הזה ".." תוספת זו מחפשת את הטקסט 

 .מעין זו הייתה חוסמת את ההתקפה שביצענו הרגע. בנתיב
 
 
 

EOF 


